
*******************************************BIS******************************* 
Source: Bank for International Settlements 
Base year = 2010 
Real effective exchange rate index 
Broad indices = 61 economies 
Narrow indices = 27 economies 
****************************************************************************; 
 
rsubmit; 
%let rolling_period=60; *cummulative return period; 
%let estimation_period=60;*time-series period used to determine whether a 
firm is exposed; 
 
**************************************CRSP variables************************; 
*Keep necessary CRSP variables; 
proc sql; 
create table a01 
as select a.permno, a.date, a.ret, b.exchcd, b.shrcd, c.vwretd, c.ewretd 
from crsp.msf as a 
left join crsp.mseall as b 
on a.permno=b.permno and a.date=b.date 
left join crsp.msi as c 
on a.date=c.date; 
quit; 
proc sort data=a01; by permno date; run; 
 
*Clean data; 
data a02; 
set a01; 
by permno date; 
retain lexchcd lshrcd; 
if first.permno then do;lexchcd = exchcd;lshrcd  = shrcd; end; 
else do; 
if missing(exchcd) then exchcd = lexchcd; else lexchcd = exchcd; 
if missing(shrcd)  then shrcd = lshrcd; else lshrcd = shrcd; 
end; 
if date>='01Jan1984'd; 
run; 
proc sort data=a02 nodupkey; by permno date ret; run; 
 
******************************************FXI return calculations***********; 
*Upload BIS data to WRDS; 
PROC upload data=exp.fxi; 
run; 
 
*create cummulative FXI returns; 
proc sql;  
create table fxi_cum 
as select distinct a.*, exp(sum(log(1+b.ret_nFXI))) - 1 as cum_nFXI, 
exp(sum(log(1+b.ret_bFXI))) - 1 as cum_bFXI 
from fxi as a, fxi as b 
where 0<=intck('month', b.date, a.date)<&rolling_period 
group by a.date having count(b.ret_nFXI)=&rolling_period; 
quit; 
 
data fxi_cum; 
set fxi_cum; 



format date YYMMDDN8.; 
informat date 8.; 
run; 
 
********************************CRSP return calculations********************; 
proc sql; *buy-hold cummulative ret; 
create table a03 
as select a.permno, a.date as cluster "Cluster", b.date, b.ret, b.vwretd, 
b.ewretd 
from a02 as a left join a02 as b 
on a.permno=b.permno and 0<=intck('month', b.date, a.date)<&rolling_period 
group by a.permno, a.date having count(b.ret)=&rolling_period and 
count(b.vwretd)=&rolling_period and count(b.ewretd)=&rolling_period 
order by a.permno, a.date, b.date; 
quit; 
 
proc sql;  
create table a04 
as select distinct permno, cluster, exp(sum(log(1+ret))) - 1 as cum_ret, 
exp(sum(log(1+vwretd))) - 1 as cum_vwretd, exp(sum(log(1+ewretd))) - 1 as 
cum_ewretd 
from a03 
group by permno, cluster 
order by permno, cluster; 
quit; 
 
proc sql; *estimation period; 
create table a05 
as select distinct a.permno, a.cluster, b.cluster as datadate 'datadate', 
b.cum_ret, b.cum_vwretd, b.cum_ewretd 
from a04 as a, a04 as b 
where a.permno=b.permno and 0<=intck('month', b.cluster, 
a.cluster)<&estimation_period 
group by a.permno, a.cluster having count(b.cum_ret)=&estimation_period 
order by a.permno, a.cluster, b.cluster; 
quit; 
 
*keep clustered data only for fiscal year ends (full Compustat sample); 
PROC upload data=fx.Comp03_all; 
run; 
PROC SQL; 
CREATE TABLE a06 AS 
SELECT a.gvkey, a.permno, a.datadate, b.cluster, b.datadate as ret_date, 
b.cum_ret, b.cum_vwretd, b.cum_ewretd, c.date as FXret_date, c.cum_nFXI, 
c.cum_bFXI 
FROM Comp03_all as a left join a05 as b 
on a.permno = b.permno and -10<=intck('day', b.cluster, a.datadate)<=10 
left join fxi_cum as c 
on -10<=intck('day', c.date, b.datadate)<=10 
order by a.permno, a.datadate, b.datadate; 
QUIT; 
 
data a06; 
set a06; 
if not missing (cluster); 
if not missing (permno); 
run; 



 
*Estimate stock return volatility to FX risk (narrow); 
%let lags=3; 
ods output parameterestimates=est_narrow; 
ods listing close; 
proc model data=A06; 
by permno datadate; 
instruments cum_vwretd cum_nFXI; 
parms b0 b1 b2; 
cum_ret=b0+b1*cum_vwretd+b2*cum_nFXI; 
fit cum_ret / gmm kernel=(bart,%eval(&lags+1),0) vardef=n; run; 
quit; 
ods listing; 
 
*Keep coefficients; 
data est_narrow1; 
set est_narrow; 
if parameter='b2'; 
drop EstType StdErr tValue DF; 
run; 
 
*Change output file name; 
proc download data = est_narrow1 out=exp.est_narrow1; 
run; 
 
*Estimate stock return volatility to FX risk (broad); 
ods output parameterestimates=est_broad; 
ods listing close; 
proc model data=A06; 
by permno datadate; 
instruments cum_vwretd cum_bFXI; 
parms b0 b1 b2; 
cum_ret=b0+b1*cum_vwretd+b2*cum_bFXI; 
fit cum_ret / gmm kernel=(bart,%eval(&lags+1),0) vardef=n; run; 
quit; 
ods listing; 
 
*Keep coefficients; 
data est_broad1; 
set est_broad; 
if parameter='b2'; 
exposed=net_negative+net_positive; 
drop EstType StdErr tValue DF; 
run; 
proc download data = est_broad1 out=exp.est_broad1; 
run; 
 
*Create clusters to estimate beta_FTR; 
proc sql; 
create table firm1 
as select a.gvkey, a.datadate as cluster "Cluster", b.fyear, b.avgfros, 
b.avgros, b.avgftr 
from Comp03_all as a left join Comp03_all as b 
on a.gvkey=b.gvkey and 0<=intck('year', b.datadate, a.datadate)<5 
group by a.gvkey, a.datadate having count(b.avgftr)=5 
order by a.gvkey, a.datadate, b.fyear; 
quit; 



 
*Estimate income shifting intensity; 
ods output parameterestimates=est_ftr; 
ods listing close; 
proc model data=firm1; 
 by gvkey cluster; 
 instruments avgros avgftr; 
 parms b0 b1 b2; 
 avgfros=b0+b1*avgros+b2*avgftr; 
 fit avgfros; 
run; 
quit; 
ods listing; 
data est_ftr2; 
set est_ftr; 
if parameter='b2'; 
if tValue ne .; 
run; 
proc download data = est_ftr2 out=exp.est_ftr2; 
run; 
endrsubmit; 
 
*Local merge; 
proc sql; 
create table exp.B1 
as select a.*, b.estimate as b_ftr, abs(c.estimate) as abs_est_narrow, 
abs(d.estimate) as abs_est_broad, 
   e.hedge 
from fx.Comp03_all as a left join exp.est_ftr2 as b 
on a.gvkey = b.gvkey and a.datadate=b.cluster 
left join exp.est_narrow1 as c 
on a.permno = c.permno and a.datadate=c.datadate 
left join exp.est_broad1 as d 
on a.permno = d.permno and a.datadate=d.datadate 
left join exp.hedge as e 
on a.ciknumber = e.ciknumber and a.fyear=e.fyear; 
quit; 
 
*Create controls and clean data; 
data exp.b2; 
set exp.b1; 
if not missing (b_ftr); 
if unique_link=1; 
size=log(1+at); 
fsale_sale=fsale/sale; 
if avgftr<0; 
if fyear<=2017; 
if not missing(abs_est_narrow+abs_est_broad+hedge); 
keep gvkey ciknumber datadate fyear sic2 abs_est_narrow abs_est_broad b_ftr 
size fsale_sale hedge; 
run; 
 
*Winsorize continuous variables here and then export; 


